

WEICDF**PURPOSE**

Compute the standard form for the Weibull cumulative distribution function with tail length parameter γ .

DESCRIPTION

For the minimum order statistic, the standard form of the Weibull cumulative distribution function is:

$$F(x) = 1 - e^{-(x^\gamma)} \quad \text{for } x \geq 0 \quad (\text{EQ 8-354})$$

For the maximum order statistic, the standard form of the Weibull cumulative distribution function is:

$$F(x) = e^{-((-x)^\gamma)} \quad \text{for } x \leq 0 \quad (\text{EQ 8-355})$$

SYNTAX

LET <y2> = WEICDF(<y1>,<gamma>) <SUBSET/EXCEPT/FOR qualification>

where <y1> is a variable, a number, or a parameter;

<y2> is a variable or a parameter (depending on what <y1> is) where the computed Weibull cdf value is stored;

<gamma> is a positive number or parameter that specifies the tail length parameter;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

EXAMPLES

LET A = WEICDF(3,2)

LET Y = WEICDF(X1,4)

NOTE 1

The SET MINMAX command specifies whether the minimum or the maximum order statistic form is used. Entering SET MINMAX 2 specifies the maximum order statistic while SET MINMAX 1 specifies the minimum order statistic.

NOTE 2

For the minimum order statistic, the general form of the Weibull cumulative distribution function is:

$$F(x) = 1 - e^{-\left(\frac{x-\mu}{\alpha}\right)^\gamma} \quad \text{for } x \geq \mu \quad (\text{EQ 8-356})$$

For the maximum order statistic, the general form of the Weibull cumulative distribution function is:

$$F(x) = e^{-\left(\frac{(-x)-\mu}{\alpha}\right)^\gamma} \quad \text{for } x \leq \mu \quad (\text{EQ 8-357})$$

where μ is the location parameter and α is the scale parameter. See topic (3) under the General considerations section at the beginning of this chapter for a discussion of generating cdf values for the general form of the distribution.

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

WEIPDF	=	Compute the Weibull probability density function.
WEIPPF	=	Compute the Weibull percent point function.
EXPCDF	=	Compute the exponential cumulative distribution function.
EXPPDF	=	Compute the exponential probability density function.
EXPPPF	=	Compute the exponential percent point function.
EVICDF	=	Compute the extreme value type I cumulative distribution function.
EVI PDF	=	Compute the extreme value type I probability density function.
EVI PPF	=	Compute the extreme value type I percent point function.

REFERENCE

“Continuous Univariate Distributions - 1,” Johnson and Kotz, Houghton Mifflin, 1970 (chapter 20).

“Statistical Distributions,” 2nd Ed., Evans, Hastings, and Peacock, Wiley and Sons, 1993 (chapter 41).

APPLICATIONS

Reliability Analysis

IMPLEMENTATION DATE

Pre-1987 (the SET MINMAX capability was implemented 93/7, earlier versions supported the minimum order statistic only)

PROGRAM

```
TITLE WEICDF FOR VARIOUS VALUES OF GAMMA
X1LABEL X; Y1LABEL PROBABILITY
SEGMENT 1 COORDINATES 69 38 74 38; SEGMENT 1 PATTERN SOLID
SEGMENT 2 COORDINATES 69 34 74 34; SEGMENT 2 PATTERN DASH
SEGMENT 3 COORDINATES 69 30 74 30; SEGMENT 3 PATTERN DOT
SEGMENT 4 COORDINATES 69 26 74 26; SEGMENT 4 PATTERN DA2
LEGEND 1 GAMMA = 1; LEGEND 1 COORDINATES 75 37
LEGEND 2 GAMMA = 2; LEGEND 2 COORDINATES 75 33
LEGEND 3 GAMMA = 5; LEGEND 3 COORDINATES 75 29
LEGEND 4 GAMMA = .5; LEGEND 4 COORDINATES 75 25
YLIMITS 0 1; MAJOR YTIC NUMBER 6
MINOR YTIC NUMBER 1; YTIC DECIMAL 1
XLIMITS 0 3
LINES SOLID DASH DOT DASH2
SET MINMAX 1
PLOT WEICDF(X,1) FOR X = 0.01 0.01 3 AND
PLOT WEICDF(X,2) FOR X = 0.01 0.01 3 AND
PLOT WEICDF(X,5) FOR X = 0.4 0.01 1.6 AND
PLOT WEICDF(X,0.5) FOR X = 0.05 0.01 2
```

